
OVERVIEW



This presentation describes the Vision system alarm package, focusing on methods and the information required to configure the alarm hierarchy and to describing two different approaches to disabling alarms, either on a group basis or on an individual basis.  



INTODUCTION



Alarms are a distributed control system's way of communicating abnormal conditions and status information to operations personal on an exception basis, when the measurement is outside or beyond pre-determined/normal limits.  Alarms are displayed on operator consoles for immediate action by operators and typically also hard copied to an alarm printer for historical purposes.



An important feature of any distributed control system is the ability to direct certain alarms to specified operator consoles in order to separate consoles according to process area responsibilities.  The Vision system accomplishes this by the assignment of each alarm (via it's associated tag) to a specific alarm group, referred to as an alarm group node name or simply group name.  These groups are arranged in a user defined hierarchical organization, which means a top down branching arrangement.  The position on the tree (node location) is determined numerically, like a telephone # (area code\exchange\phone#).  This number is called the group number and is given a name which is typically the group name with a "_G" suffix.





GROUP NODE NAME: TK1234	  group name

GROUP TYPE(def: LOGICAL):	  all process alarms would use logical

MASK INDEX: 2			  this defines the node level (1=top)

LOGGING CUTOFF(def: 0):	  not implemented (for disk logging)

NUMBER OF GROUPS: 1		  allows multiple group #'s to share 1 node

GROUP NUMBER: TK1234_G	  symbol name of group number value

FIELDS DEFINITION: 3\1\0\0  4 byte numeric value for node definition

etc...



Each alarm group definition contains the declaration for any number of operator consoles and a hard copy "alarm printer".  The alarms are then "sent" or routed only to those consoles and printers (collectively called "I/O UNITS") specified for that group.  Each video or print unit assigned to a group can also be further configured to restrict certain aspects of the alarm display or control access privilege.  The assignment of I/O units to a group applies only to that group and has no impact on the groups below it.



In addition to the "routing" task performed by the alarm group configuration, there is also the ability to mask or suppress alarms, based on group name.  This can be done dynamically, with an algorithm, or more permanently in a BLOC builder dialog.  The disabling of a group node will also suppress the alarms from all descendent nodes. 

ALARM HEIRARCHY BACKGROUND



The alarm organization is commonly referred to as the alarm "tree" since it branches downward from the root node "PROC_ALM" for the process alarms or "SYS_ALM" for the system (maintenance) alarms.  A groups position on the tree is determined by the numeric value in the "FIELDS DEFINITION" (following the "GROUP NUMBER" prompt).  This group number is in fact a named symbol typically made up of 4 numbers, each from 0 to 255 (other combinations are possible but difficult to setup).  



�

�ALARM & TAG ASSIGNMENT

Most alarms are automatically created as part of the tagname generation process and are by definition associated with that tagname.  That tagname has the "link" to the alarm group name, by an entry in the loop table array.  

The user can however defines his own "custom" alarms (via the "SENDALARM" or "SENDALRM0" BLOC algorithms). He would need either to link them with an existing tag or create a new tag which points to the desired group.  A macro is illustrated later which simplify this process.



�

   The tag's loop table array points to the alarm group name which provides the logical "grouping" alarms for different process areas.  

�The group's definition in turn contains the information which logically positions the group on the hierarchy (which is typically 4 levels deep) and "attaches" or assigns operator consoles and printer units.  Shown below is the group definition for group "TK1234" which has 3 op consoles and one serial alarm printer assigned.



�



It should be pointed out that the Vision system global data base has a finite capacity for storing and supporting alarm group nodes and I/O units.  Due to space limitations in how the Vision system stores the group configuration data in it's bank memory, the system can appear somewhat limited in its capacity.  Estimates for alarm group storage is that for groups which each have 4 I/O units, the system can support about 100 - 120 groups.  However, on a system which has about 7 I/O units per group that number dropped to 65 group capacity.  This might not be as much of a concern if groups could "inherit" the I/O information from it's parent nodes, but that is not yet the case.



So the user should exercise caution when setting up the alarm group planning and configuration phases, not to exceed the systems alarm group capacity.



VIDEO AND PRINTER CONFIGURATION



The topic of operator console and printer configuration was covered in the May 1989 (pg18) issue of the "Vision Perspectives", but a quick review would be helpful.  The most important issue is the understanding of the term "MLUN" (Master Logical Unit Number).  This is simply the term used to refer to the sequential numbering of all I/O Units (op consoles, graphic printers, & serial printers) in a global network.  The number is must be unique for each type of unit.  Generally the first unit is #1, 2nd #2, etc...  

�To configure operator consoles or graphic printers the user can either use the I/O Config tool (within the "port" subject) or use the BLOC Builder macros.  For serial alarm printers and the console device for the MX-63 board the user is limited to the BLOC Builder macros.  The macros names, description and their parameters are listed for reference.



BLOC BUILDER MACROS FOR VIDEO AND PRINTER CONFIGURATION





CFG_VIDEO.ATR - Configures an operator console.

Parameters:

1.)  Video Board #: the leftmost board is #4 and used 1st, then counts down to #1.

2.) BSC port for button: starts counting at #1 and goes up. Typically 1 - 4.

3.) MLUN for graphics printer for the hard copies from this op console.

4.) LPN name where this op console resides.

5.) MLUN assigned to this operator console. Must be unique for all op cons 

		within the global network.

6.) Baud Rate: 8 for 9600; 16 for 19200



CFG_PRINT.ATR - Configures a graphics printer.

Parameters:

1.) MLUN assigned to this graphic printer. Must be unique for all graphic 

		printers within the global network.

2.) Printer Board #: the leftmost board is #5 and rightmost #6.

3.) LPN name where this printer resides.



CFG_SERIAL.ATR - Configures a serial printer (connected to a BSC port).

1.) LPN name where this printer resides.

2.) LPN#: the number portion of the LPN name given for #1.

3.) MLUN assigned to this serial printer. Must be unique for all serial 

		printers within the global network.

4.) BSC Port #: depends on hardware assignment. Between 1 - 12 possible.



CFG_CONSOLE.ATR - Configures a the MX60/63 console/terminal port.

1.) LPN name where this terminal resides.

2.) LPN#: the number portion of the LPN name given for #1.





Once the consoles and printers are created with the above macros, then can then be configured as part of the alarm routing process from the alarm groups.

�ALARM GROUP CONFIGURATION & UNIT ASSIGNMENT



The following show the creation of a new group with one video unit being added at this time.  This video is configured not to show alarms below a priority of 10 (cutoff code).  The reason for the cutoff code of 10 is to allow certain alarms to not appear on the console, but still appear on the printer.  The entry in the FLAGS field of "(ACK,CONTROL)" allows acknowledgement of the groups alarms and allows tags belonging to this group to be "called" onto the tagline and control actions then permitted.  





GROUP NODE NAME: TK1234	  group name

GROUP TYPE(def: LOGICAL):	  all process alarms would use logical

MASK INDEX: 2			  this defines the node level (1=top)

LOGGING CUTOFF(def: 0):	  not implemented (for disk logging)

NUMBER OF GROUPS: 1		  allows multiple group #'s to share 1 node

GROUP NUMBER: tk1234_G	  symbol name of group number value

FIELDS DEFINITION: 3\1\0\0	  4 byte numeric value for node definition

NUMBER OF UNITS(def: 1):	  define 1st I/O unit to assign to node

 SYSLPN(def: lpn4): LPN4	  LPN of executive for this unit

 IOTYPE: VIDEO_DSPLY		  choices VIDEO_DSPLY, ALM_PRINT, FRAMCOPY

 MLUN: 3				  Master Logical Unit # for this video

 CUTOFF CODE: 10		  don't display alarms with priority less than

 UNIT TYPE(def: VIDEO):	  type of I/O unit

 FLAGS: (ACK,CONTROL)	  allows acknowledgement and control line access

 DELTA CODE: 0			  adds (or subtracts) this to alarms priority

 BRANCH FRAME 1(def: 0):	  not used feature

 BRANCH FRAME 2(def: 0):	  not used feature

 BRANCH FRAME 3(def: 0):	  not used feature

 BRANCH FRAME 4(def: 0):	  not used feature





The following shows the assignment of a new printer unit to an existing group. This uses the modify group with the "AUNIT" (for Assign Unit) option.  The unit being added is the serial alarm printer (MLUN #1) connected to a BSC port.  This printer's setting of FLAGS to "(PRACK,PRCLEAR)" will print both the alarm acknowledgements and alarm clearing as well as the alarm occurrences.





MODIFY GROUP

GROUP NODE NAME: TK1234

MODIFICATION TYPE: HELP

INPUT IS ONE OF THE FOLLOWING: 'DATA', 'AGROUP', 'DGROUP', 'AUNIT', 'DUNIT'

MODIFICATION TYPE: AUNIT

NUMBER OF UNITS (DEF: 1)(cur: 1):

 SYSLPN(cur: lpn4):

 IOTYPE: ALM_PRINT

 MLUN: 1

 CUTOFF CODE: 0

 UNIT TYPE (DEF: VIDEO): HELP

INPUT IS ONE OF THE FOLLOWING KEYWORDS: 'VIDEO' AND 'PRINT'.

 UNIT TYPE (DEF: VIDEO): PRINT

 FLAGS: (PRACK,PRCLEAR)





�The following shows the assignment of a graphics printer unit to an existing alarm group.  Not many people use this option, since it clutters the hard copy graphics printer with alarm messages, but it is sometimes useful to have the ability to configure a system this way.





MODIFY GROUP

GROUP NODE NAME: TK1234

MODIFICATION TYPE: AUNIT

NUMBER OF UNITS (DEF: 1)(cur: 1):

 SYSLPN(cur: lpn4):

 IOTYPE: FRAMCOPY

 MLUN: 1

 CUTOFF CODE: 0

 UNIT TYPE (DEF: VIDEO): PRINT

 FLAGS: (PRACK,PRCLEAR)





The following shows the resulting group with all three units listed.





BLOC: LIS GRO TK1234

GROUP NODE NAME: tk1234

GROUP TREE TYPE: LOGICAL

PREDECESSOR NODE: proc_alm

MASK INDEX: 2

ACTIVE DISABLE: "NONE"

LOGGING CUTOFF CODE: 0

GROUP NUMBER 1: tk1234_g, 3\1\0\0

UNIT(S) DATA-



UNIT: lpn4 - video_dsply - 61

TYPE: VIDEO

CUTOFF CODE: 0

FLAGS ON: "ACK","CONTROL"

STATUS: LOCALLY ENABLED

DELTA CODE: 0

BRANCH FRAME 1: 0

BRANCH FRAME 2: 0

BRANCH FRAME 3: 0

BRANCH FRAME 4: 0



UNIT: lpn4 - alm_print - 1

TYPE: PRINT

CUTOFF CODE: 0

FLAGS ON: "PRACK","PRCLEAR"

STATUS: LOCALLY ENABLED



UNIT: lpn4 - framcopy - 1

TYPE: PRINT

CUTOFF CODE: 0

FLAGS ON: "PRACK","PRCLEAR"

STATUS: LOCALLY ENABLED





�

ALARM PRIORITY ASSIGNMENT



No discussion of alarms would be complete without a mention of alarm priority's.  It  is quite simply a method of color differentiation and acknowledgement behavior.  One significant aspect is that on the PIDP boards that the priority can be individually assigned for each alarm within each loop (CCP and AIP board had only one setting per tag and depended on the default setting for color differentiation).  The configuration for PIDP loop's priority's are shown below.



Listing of a PID loop's device, showing the location and usage of each 

alarm's priority specification for the alarms associated with a PID loop.

Note that a zero entry specifies the"default"value, not a zero value.



BLOC: lis dev FC1234dio



    IOMASTER: FC1234dio

    COMMENT:

    TYPE: pidp_ddc_ctl               BOARD: pid0601

    LPN: lpn1                        LOGICAL PORT : port06

    BTI: btip1                       BOARD ADDRESS: 1

    CONFIG. DATA LOCATION: BULKRES   PHYSICAL PORT: 6



    DEVICE DESCRIPTOR DATA -

      dev_number: 1

      dev_type: 71

      looptbladdr: lpn1\btidp1\FC1234

      alarm1_prty: 0    !(for Rate of Change Alarm)

      alarm2_prty: 0    !(for Low       (Soft Low)  Alarm priority)

      alarm3_prty: 0    !(for High      (Soft High) Alarm priority)

      alarm4_prty: 0    !(for Low Low   (Hard Low)  Alarm priority)

      alarm5_prty: 0    !(for High High (Hard High) Alarm Priority)

      alarm6_prty: 0    !(for Error Deviation (SP-PV) Alarm Priority)

      alarm7_prty: 5    !(for SetPt High Alarm Priority)

      alarm8_prty: 5    !(for SetPt Low  Alarm Priority)

      alarm9_prty: 0    !(for Pr Output at Limit Alarm Priority)

        .

         etc...



     (Note: The comments fields are added manually and will not appear in BLOC Builder.)



It is user selectable in BLOC Builder (in the PARAMETER ALARM subject) at what thresholds the alarm color and behavior changes, however most user do not change these settings from the factory defaults.  These are:



HIGH HIGH PRIORITY = 192 -255 (will not clear screen until acknowledged)

HIGH PRIORITY              = 128 - 191(will not clear screen until acknowledged)

MEDIUM PRIORITY       =   64 - 127(will clear screen when condition clears)

LOW PRIORITY                 =     1 - 63 (will not blink appears already acknowledged)

�Custom Alarm Creation 



The previous discussion has focused on standard alarms, created automatically when a tagname is generated.  It is quite simple to create custom alarms to suit particular needs, the simplest is the need of a contact input to generate an alarm.  This can be done by a user configuring the algorithm in BLOC, but it is much easier to leave the repetitive part of the task to a macro and minimize ones typing work.



The macro shown below will perform the task of adding a sendalrm0 algorithm to a block, along with the housekeeping task of defining the arrays for pointing to the group and also insuring that the names have not been duplicated.



FILE: ALARM.BBM

.macro

.prompt  "P1, Enter a UNIQUE label for the alarm (9 Chars Max)"

.prompt  "P2, Enter the already defined ordinal to trigger the alarm"

.prompt  "P3, Enter the alarm severity (0 - 255)"

.default "P3, 50"

.prompt  "P4, Enter the already defined alarm group name"

.prompt  "P5, Enter the Description field (36 chars Max)"

.prompt  "P6, Enter the name of the LPN (ie: LPN1)"

.default "P6, LPN1"

.prompt  "P7, Enter the name of the Chain (ie: CHN.ALARM)"

.default "P7, CHN.ALARM"

.prompt  "P8, Enter the name of the Block (ie: BLK.ALARM)"

.default "P8, BLK.ALARM"

.prompt  "P9, Enter the BTI (ie: BTID1)"

.default "P9, BTID1"



! This macro will add a sendalrm0 algorithm to an existing user 

! defined block (which should be rate based).  The macro assumes that

! the alarm group number ID is the entered group name with "_G" suffix.



! This macro will also add the necessary loop table array to support

! the algorithms requirement for the pointer to the group node name.

! One loop table will be added once per different group reference, so

! subsequent calls will produce the harmless "array already defined"

! error messages (the "set def error continue" flag allows the audit 

! trail to continue).



! The macro call is: 

! @ALARM.BBM,'%P1','%P2','%P3','%P4','%P5','%P6','%P7','%P8','%P9'



SET DEF LPN '%P6'

SET DEF BTI '%P9'

SET DEF ERR STOP

LIS SYM '%P2'

LIS SYM '%P4'_G

LIS SYM '%P8'



ADD SYM '%P1'SEV = '%P3',(VAR,BYTE),'%P6'\'%P9'



SET DEF ERR CONT



ADD ARRAY '%P6'\'%P9'\'%P4'DT,DATENTDTTBL,"DATA TBL FOR GROUP '%P4'",-

0.0,1000000.0,-1000000.0



ADD ARRAY '%P6'\'%P9'\'%P4'.G,DATENTTBL,"LOOP TBL FOR GROUP '%P4'",-

'%P4'_G,1,NULL,NULL,'%P6'\'%P9'\'%P4'DT,"'%P4'","GROUP"," "



SET DEF ERR STOP



ADD ALG '%P6'\'%P7'\'%P8',,SENDALRM0,-

'%P2',ALARM0,'%P4'.G,-

'%P4'.G\GROUPNMBR,-

'%P6'\'%P9'\'%P1'SEV,-

('%P6'\'%P9'\'%P1'LON,Y,"- '%P5'"),-

('%P6'\'%P9'\'%P1'SHO,Y,"'%P1'"),-

('%P6'\'%P9'\'%P1'ATG,Y,"'%P1'"),-

('%P6'\'%P9'\'%P1'OLD,Y,0)





The macro is also available in the sendalarm version which supports the display of a user defined floating point value with a shorter text field (18 chars).  This is useful for alarms with associated measurements or calculated values not covered by Unitec alarms.



Alarm Group Disabling



Alarm group disabling, described earlier, can be done dynamically, with an algorithm, or more permanently (requiring a reboot to disable or to restore), in a BLOC builder dialog.  Group disabling is performed most efficiently using the algorithm "setalarm".  The BLOC Builder method of alarm group disabling is considered to be of limited usefulness, since the system must be rebooted to activate the change and that it must be performed by a user skilled with BLOC Builder.



The SETALARM algorithm, on the other hand, has the advantage of being able to permit or suppress alarms based on an ordinal state, which would be set or cleared based on logic in anticipation of the need to suppress alarms (for a shutdown, for instance).  The key word is anticipation, since the setalarm algorithm does not operate retroactively, in either direction.  This means that it must be triggered before the need to suppress, but more importantly it must be restored before alarms need to be seen again.  This is limitation must be borne in mind when using this feature.  



One general consideration related to alarm disabling or suppressing is that it remarkably easy to forget or not realize that the alarm have been disabled unless there is some obvious indicator that a disable is active.  This need has been addressed in the following macro.  The approach is to disable the group using an ordinal while at the same time that ordinal is used to generate a low priority (white colored) alarm which appears on a selected console as a reminder.  The logic has been written to support the hiding or displaying of these white disabled status alarms by the use of a disabled status display ordinal ("DISABL_DISP").  This way one can, if desired even hide these alarms if they are a nuisance.

�FILE: DISABLED.BBM

.macro

.prompt  "P1, Enter the already defined alarm group name"

.prompt  "P2, Enter the LPN name of the executive (ie: LPN1)"

.default "P2, LPN1"

.prompt  "P3, Enter the BTI (ie: BTID1)"

.default "P3, BTID1"

.prompt  "P4, Enter the unit # (MLUN) of video for message (ie: 1)"

.default "P4, 1"



! This macro will disable the alarm group specified by the user

! based on the user set/cleared ordinal state (ON = disable).

! The group "'%P1'" will be disabled by ordinal "'%P1'AD"



SET DEF ERR CONT



ADD CHAIN '%P2'\CHN.ALARM,,bct1sec,btib1,6,1,1,-

(COLD,TEPID),'%P3','%P2'\'%P3'\$l.on

ADD BLOCK '%P2'\CHN.ALARM\BLK.ALARM,,1,'%P2'\'%P3'\$l.on,ACTIVE



ADD GROUP DISABLED,LOGICAL,1,0,1,(DISABLED_G,254\0\0\0),1,-

'%P2',video_dsply,'%P4',0,VIDEO,ACK,0,0,0,0,0



ADD SYMBOL DISABL_DISP=ON,ORD,'%P2'\'%P3'



SET DEF ERR STOP

ADD SYMBOL '%P1'AD=OFF,ORD,'%P2'\'%P3'

ADD SYMBOL '%P1'DD=OFF,ORD,'%P2'\'%P3'



! DISPITE LINE WRAP, THE FOLLOWING CALL TO ALARM.BBM IS ONE LONG LINE

@ALARM.BBM,'%P1'_D,'%P1'DD,50,DISABLED,ALARM GROUP '%P1' DISABLED,'%P2',CHN.ALARM,BLK.ALARM,'%P3'



ADD ALGORITHM '%P2'\chn.alarm\blk.alarm,,logic,"UNIVERSAL",-

"IC",'%P2'\'%P3'\'%P1'AD,-

"XAC",'%P2'\'%P3'\DISABL_DISP,-

"OC",'%P2'\'%P3'\'%P1'DD,END



ADD ALG '%P2'\CHN.ALARM\BLK.ALARM,logic,setalarm,-

'%P1'AD,"A_GLOBAL",video_dsply,-

('%P2'\'%P3'\mlun'%P4',Y,'%P4'),-

('%P2'\'%P3'\'%P1'ID,Y,"'%P1'"),-

('%P2'\'%P3'\'%P1'_H,Y,0)





�Individual Alarm Disabling



There are times when it is desirable to disable individual alarms or alternatively the need to make the disabling of alarms work retroactively is paramount.  An approach for accomplishing both of these requirements is shown below.  It is coded on the PIDP program device and has the advantage of being able to efficiently disable alarms after they have occurred or conversely will trigger alarms which went true while disabled.  The drawback is obviously the significant amount of code for each loop needing to be treated in this manor, plus the requirement that the user must transmit the disable flag from the VPU to the PIDP flag "DISABL_ALM" to activate the logic.



PIDP PROGRAM DEVICE

Local_Boolean

   DISABL_ALM;                ! DISABLE FLAG FOR ALARM, SET BY LOGIC

External

   FC1234DIO\HHLIMITVAL,	! HIGH HIGH VALUE

   FC1234DIO\HSLIMITVAL,	! HIGH VALUE

   FC1234DIO\LSLIMITVAL,	! LOW LIMIT VALUE

   FC1234DIO\LHLIMITVAL;	! LOW LOW LIMIT VALUE



!**********************************************

!*         Alarm Disabling Code               *

!**********************************************



! HIGH HARD disable if DISABL_ALM

IF DISABL_ALM and FC1234DIO\HHLIMITVAL LT 1000.0 Then

   FC1234DIO\HHLIMITVAL = FC1234DIO\HHLIMITVAL + 2000.0;

EndIf;

If NOT DISABL_ALM AND FC1234DIO\HHLIMITVAL GT 1000.0 Then

   FC1234DIO\HHLIMITVAL = FC1234DIO\HHLIMITVAL - 2000.0;

EndIf;



! HIGH SOFT disable if DISABL_ALM

IF DISABL_ALM and FC1234DIO\HSLIMITVAL LT 1000.0 Then

   FC1234DIO\HSLIMITVAL = FC1234DIO\HSLIMITVAL + 2000.0;

EndIf;

If NOT DISABL_ALM AND FC1234DIO\HSLIMITVAL GT 1000.0 Then

   FC1234DIO\HSLIMITVAL = FC1234DIO\HSLIMITVAL - 2000.0;

EndIf;



! LO SOFT disable if DISABL_ALM

IF DISABL_ALM AND FC1234DIO\LSLIMITVAL GT -1000.0 THEN

   FC1234DIO\LSLIMITVAL = FC1234DIO\LSLIMITVAL - 2000.0;

ENDIF;

IF NOT DISABL_ALM AND FC1234DIO\LSLIMITVAL LT -1000.0 THEN

   FC1234DIO\LSLIMITVAL = FC1234DIO\LSLIMITVAL + 2000.0;

ENDIF;



! LO HARD disable if DISABL_ALM

IF DISABL_ALM AND FC1234DIO\LHLIMITVAL GT -1000.0 THEN

   FC1234DIO\LHLIMITVAL = FC1234DIO\LHLIMITVAL - 2000.0;

ENDIF;

IF NOT DISABL_ALM AND FC1234DIO\LHLIMITVAL LT -1000.0 THEN

   FC1234DIO\LHLIMITVAL = FC1234DIO\LHLIMITVAL + 2000.0;

ENDIF;

�But even with these drawbacks, this logic has proved very useful in solving the major problem with the setalarm approach of alarm disabling, the retroactively "waking up" of past missed alarms or the purging from the screen alarms currently active but recently disabled.  Macros make the replication an easy task.  



Although not as important a requirement, contact input disabling could be accomplished with a logic algorithm just ahead of the sendalrm0 algorithm.  It would use a group disable ordinal to mask a new ordinal which would trigger the alarm.  A code fragment is shown below:



LOGIC FOR CONTACT INPUT ALARM DISABLING

ADD ALGORITHM lpn4\chn.alarm\blk.alarm,,LOGIC,"UNIVERSAL",-

"IC",CI1234LS,-

"XAC",TK1234AD,-

"OC",CI1234AL,END



ADD ALG lpn4\chn.alarm\blk.alarm,LOGIC,SENDALRM0,-

CI1234AL,ALARM0,TK1234.G,-

TK1234.G\GROUPNMBR,-

(LPN4\BTID1\CI1234SEV,-

(LPN4\BTID1\CI1234LON,Y,"- CONTACT INPUT ALARM ACTIVE, HELP!!"),-

(LPN4\BTID1\CI1234SHO,Y,"CI1234"),-

(LPN4\BTID1\CI1234ATG,Y,"CI1234"),-

(LPN4\BTID1\CI1234OLD,Y,0)



This might be useful if it is desired to have the convenience of purging and refreshing past alarms whenever the disable was set or cleared or if it were important to single out individual alarms for disabling.  Again remember how easy it could be to be depending on an alarm which was accidently left disabled.  Be sure to provide an obvious way to see the disabled status, such as shown in the macro above ("DISABLED.BBM") or perhaps on a video display created for the purpose.



Summary



So during this discussion, we have examined the alarm package and highlighted various aspects of group related configuration data and the interrelationship of tags, alarms groups, and I/O units in the Vision system.  We have examined methods for simple custom alarm generation and several methods for disabling alarms when they are not desirable to be displayed. 

